# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Jan Magnus Brevik <janmbrevik@gmail.com>

_pkgname=podcastparser
pkgname=python-podcastparser
pkgver=0.6.10
pkgrel=2
pkgdesc='Simplified, fast RSS parsing library in Python'
arch=('any')
url='https://github.com/gpodder/podcastparser'
license=('ISC')
depends=('python')
checkdepends=('python-pytest-cov')
source=("https://github.com/gpodder/$_pkgname/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('b407c0b40f788e9b49e3f7f1eb1f514d452f241ec3b1020ae207d5329fec1526')

build() {
  cd $_pkgname-$pkgver
  python setup.py build
}

check() {
  cd $_pkgname-$pkgver
  pytest
}

package() {
  cd $_pkgname-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1 --skip-build
  install -Dm644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
